



Freefem++

F. Hecht

Laboratoire Jacques-Louis Lions
Université Pierre et Marie Curie
Paris, France

with A. Le Hyaric, P. Jolivet, I. Danaila, O. Pironneau

<http://www.freefem.org>

<mailto:hecht@ann.jussieu.fr>

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools
- 5 Schwarz method with overlap
- 6 Advance exercice
- 7 No Linear Problem
- 8 Technical Remark on freefem++
- 9 Phase change with Natural Convection

1 Introduction

- History
- The main characteristics
- Changes
- Basement
- Weak form
- Poisson Equation

2 Tools

3 Academic Examples

4 Numerics Tools

5 Schwarz method with overlap

Message of P. Jolivet on Curie computer at CEA : (see Tutorial to tomorrow afternoon).

Méthode de décomposition de domaine à deux niveaux pour résoudre des problèmes elliptiques très hétérogènes

- $22 \cdot 10^9$ d'inconnues en 2D (résolu en 200 secondes sur plus de 16000 coeurs)
- $0.3 \cdot 10^9$ d'inconnues en 3D (résolu en 200 secondes sur plus de 8000 coeurs).

The 6th workshop FreeFem++ : Jussieu, Paris 9,10,11 décembre, 2014.

- Morning : presentation
- Afternoon : tutorial

Introduction

FreeFem++ is a software to solve numerically partial differential equations (PDE) in \mathbb{R}^2) and in \mathbb{R}^3) with finite elements methods. We used a user language to set and control the problem. The FreeFem++ language allows for a quick specification of linear PDE's, with the variational formulation of a **linear steady state problem** and the user can write they own script to solve no linear problem and time depend problem. You can solve coupled problem or problem with moving domain or eigenvalue problem, do mesh adaptation , compute error indicator, etc ...

By the way, FreeFem++ is build to play with abstract linear, bilinear form on Finite Element Space and interpolation operator.

FreeFem++ is a freeware and this run on Mac, Unix and Window architecture, in parallel with MPI.

- 1987 MacFem/PCFem les ancêtres (O. Pironneau en Pascal) payant.
- 1992 FreeFem réécriture de C++ (P1,P0 un maillage) O. Pironneau, D. Bernardi, F. Hecht , C. Prudhomme (adaptation Maillage, bamg).
- 1996 FreeFem+ réécriture de C++ (P1,P0 plusieurs maillages) O. Pironneau, D. Bernardi, F. Hecht (algèbre de fonction).
- 1998 FreeFem++ réécriture avec autre noyau élément fini, et un autre langage utilisateur ; F. Hecht, O. Pironneau, K.Ohtsuka.
- 1999 FreeFem 3d (S. Del Pino) , Une première version de freefem en 3d avec des méthodes de domaine fictif.
- 2008 FreeFem++ v3 réécriture du noyau élément fini pour prendre en compte les cas multidimensionnels : 1d,2d,3d...

For who, for what !

For what

- 1 R&D
- 2 Academic Research ,
- 3 Teaching of FEM, PDE, Weak form and variational form
- 4 Algorithmes prototyping
- 5 Numerical experimentation
- 6 Scientific computing and Parallel computing

For who : the researcher, engineer, professor, student...

The mailing list <mailto:Freefempp@ljll.math.upmc.fr> with 410 members with a flux of 5-20 messages per day.

More than 2000 true Users (more than 1000 download / month)

- Wide range of finite elements : continuous P1,P2 elements, discontinuous P0, P1, RT0,RT1,BDM1, elements ,Edge element, vectorial element, mini-element, ...
- Automatic interpolation of data from a mesh to an other one (with matrix construction if need), so a finite element function is view as a function of (x, y, z) or as an array.
- Definition of the problem (complex or real value) with the variational form with access to the vectors and the matrix.
- Discontinuous Galerkin formulation (only in 2d to day).
- LU, Cholesky, Crout, CG, GMRES, UMFPack, SuperLU, MUMPS, HIPS , SUPERLU_DIST, PASTIX. ... sparse linear solver ; eigenvalue and eigenvector computation with ARPACK.
- Online graphics with OpenGL/GLUT/VTK, C++ like syntax.
- An integrated development environment FreeFem++-cs

- Analytic description of boundaries, with specification by the user of the intersection of boundaries in 2d.
- **Automatic mesh generator**, based on the Delaunay-Voronoi algorithm. (2d,3d (tetgen))
- load and save Mesh, solution
- **Mesh adaptation based on metric**, possibly anisotropic (only in 2d), with optional automatic computation of the metric from the Hessian of a solution. (2d,3d).
- Link with other soft : parview, gmsh , vtk, medit, gnuplot
- Dynamic linking to add plugin.
- Full MPI interface
- Nonlinear Optimisation tools : CG, **lpopt**, NLOpt, stochastic
- Wide range of examples : Navier-Stokes 3d, elasticity 3d, fluid structure, eigenvalue problem, Schwarz' domain decomposition algorithm, residual error indicator ...

The Changes form 09/11 to 06/13

- v 3.16
- cmaes interface in scalar and MPI case (thank to S. Auliac)
- add NLOpt interface (thank to S. Auliac)
- build a pkg under macos for distribution .
- rewrite the isoline-P1 plugins
- v 3.18 (11/01/2012)
- add tools for adaptation of P2 and P3 finite elements with metrics
- add plugins with sequential mumps without mpi
- add conversion of re and im part of complex sparse matrix
- add Ipopt interface (thanks to Sylvain Auliac)
- scotch partitionner interface see scotch.edp
- v 3.19 (20 april 2012)
- add tool to create Quadrature formular 1d,2d,3d with
- add integration on levelset line (in test)
- add formal tools on array [] or matrix [[],[]] for elastic problem.
- add new MUMPS parallel version plugin
- add paradiso interface (MKL)
- version 3.22
- add multi windows graphics ; WindowIndex=0 in plot function and add new event in graphic windows * to set/unset default graphics stat to previous plot
- add getenv, setenv , unsetenv function in shell plugins for the management of environemnt variable for openmp.
- correct pb un trunc for 3d mesh with too flat element (sliver) , and cleaning code .
- correct bug of the domain outside flag in 3d in case when we use the brute force (searchMethod>0)
- version 3.23
- do cleaning in version remove x11, glx, std : freefem++
- add flags to remove internal boundary in 2d,3d in function change rmInternalEdges=1
- glumesh in case of no mesh in 2d
- correct extract function of mesh Lo Sala <salalo80@gmail.com>
- correct int2d on levelset see example intlevelset.edp
- correct automake TESTING part (in progress)
- correct typo on the doc with .*= ./= operator
- correct bug in RT0 3d , code in the construction of the DOF.
- add new parameter to ffglut for demo of freefem++
- to version 3.26-2. Correct compile problème, configuration clean up , ...

The Changes form 06/13 to 12/14

- correct misstake in examples++-3d/MeshSurface.idp
- correct a bug the DG with periodic boundary condition with only one layer of element.
- add plugging "bfstream" to write and read in binary file (long, double, complex<double> and array) see bfstream.edp for an example. version 3.30-1 may/2014 (hg rev : 3017)
- add levelset integral on 3d case
- correct problem with lpopt / lapack configure ...
- add BEC plugin of Bose-Einstein Optimisation
- standardisation movemesh3 -> movemesh (same parameter of 2d version)
- correct jump in basic integral to be compatible with varf definition
- version 3.30.
- add binary ios :mode constant, to open file in binary mode under window to solve pb of seekg under windows
- add multy border april 23 2014 , (hg rev : 3004) syntaxe example :
- add ltime() (rev 2982) function returns the value of time in seconds since 0 hours, 0 minutes, 0 seconds, January 1, 1970, (int)
- add new macro tool like in C (rev 2980) FILE,LINE,Stringification() to get line number and edp filename,
- add new int2d on levelset in 3d (int test)
- correct bug in periodic condition in case common dof with periodic.
- correct big bug in memory gestion of sparse matrix
- version 3.32
- correct of problem of plugin and mpi,
- correct of plugin MUMPS.cpp for complex value.
- add vectorial operator a/v and v/a where a est scalar and v vector like real[int], ...
- correct the problem of size of arrow in 2d plot
- version 3.31-2 (rev 3052, 11 july 2014)
- correct stop test function in LinearGC (for zuqi.tang@inria.fr)
- correct bug put in DG formulation (rev 3044) jump, mean , was wrong from Sun Jun 29 22 :39 :20 2014 +0200 rev 3028 version 3.31-1 (rev 3042, 10 july 2014)
- function to put your stop test in LinearGC and NLGC the prototype is
- add fonctionnal interface to arpack (Eigen Value) see examples++-eigen/LapEigenValueFunc.edp for a true example
- version 3.31 (rev 3037, 1 july 2014)
- re-add tan function for complex number
- correct a big mistake in LinearGMRES , the resultat are complity wrong, correct also the algo.edp
- add sqr fonction of O. Pironneau
- correct update of mercurial depot (rev 3034, 1 july 2014)

The key words are reserved

The operator like in C exempt: ^ & |

+ - * / ^ // $a^b = a^b$

== != < > <= >= & | // $a|b \equiv a \text{ or } b$, $a\&b \equiv a \text{ and } b$

= += -= /= *=

BOOLEAN: 0 <=> false , \neq 0 <=> true = 1

// Automatic cast for numerical value : bool, int, reel,
// complex , so
func heavyside = real(x>0.);

for (int i=0;i<n;i++) { ...;}
if (<bool exp>) { ...; } else { ...;};
while (<bool exp>) { ...;}
break continue key words

weakless: all local variables are almost static (???)
bug if break before variable declaration in same block.
bug for fespace argument or fespace function argument

```

x,y,z , label,           //    current coord., label of BC.
N.x, N.y, N.z,           //    normal
int i = 0;                //    an integer
real a=2.5;               //    a reel
bool b=(a<3.);
real[int]  array(10);     //    a real array of 10 value
mesh Th; mesh3 Th3;       //    a 2d mesh and a 3d mesh
fespace Vh(Th,P2);        //    Def. of scalaire 2d finite element space;
fespace Vh3(Th3,P1);      //    Def. of scalaire 3d finite element
space;
Vh u=x;                   //    a finite element function or array
Vh3<complex> uc = x+ 1i *y; //    complex valued FE
u(.5,.6,.7);              //    value of FE function u at point (.5,.6,.7)
u[];                       //    the array of DoF value assoc. to FE function u
u[][5];                   //    6th value of the array (numbering begin
                           //    at 0 like in C)

```

```

fespace V3h(Th, [P2,P2,P1]);
V3h [u1,u2,p]=[x,y,z];           // a vectorial finite element
                                   // function or array
    // remark u1[] <==> u2[] <==> p[] same array of unknown.
macro div(u,v) (dx(u)+dy(v)) // EOM a macro
                                   // (like #define in C )
macro Grad(u) [dx(u),dy(u)]      // the macro end with //
varf a([u1,u2,p],[v1,v2,q])=
    int2d(Th) ( Grad(u1)'*Grad(v1) +Grad(u2)'*Grad(v2)
              -div(u1,u2)*q -div(v1,v2)*p)
    +on(1,2) (u1=g1,u2=g2);

matrix A=a(V3h,V3h,solver=UMFPACK);
real[int] b=a(0,V3h);
u2[] =A^-1*b;                     // or you can put also u1[]= or p[].
```

```

func Heaveside=(x>0);           // a formal line function
func real g(int i, real a) { .....; return i+a;}
A = A + A'; A = A'*A           // matrix operation (only 1/1)
A = [ [ A,0],[0,A'] ];        // Block matrix.
int[int] I(15),J(15);          // two array for renumbering
// the aim is to transform a matrix into a sparse matrix
matrix B;
B = A;                         // copie matrix A
B=A(I,J);                      // B(i,j) = A(I(i),J(j))
B=A(I^-1,J^-1);               // B(I(i),J(j))= A(i,j)
B.resize(10,20);              // resize the sparse matrix
// and remove out of bound terms
int[int] I(1),J(1); real[int] C(1);
[I,J,C]=A;                    // get of the sparse term of the matrix A
// (the array are resized)
A=[I,J,C];                    // set a new matrix
matrix D=[diagofA];           // set a diagonal matrix D
// from the array diagofA.
real[int] a=2:12;             // set a[i]=i+2; i=0 to 10.

```

*a formal array is [exp1, exp1, ..., expn]
 the Hermitian transposition is [exp1, exp1, ..., expn]'*

```
complex a=1,b=2,c=3i;
func va=[ a,b,c];           //      is a formal array in [ ]
a =[ 1,2,3i]'*va; cout « a « endl;           //      Hermitian product
matrix<complex> A=va*[ 1,2,3i]'; cout « A « endl;
a =[ 1,2,3i]'*va*2.;
a =(va+[ 1,2,3i])'*va*2.;
va./va;                      //      term to term /
va*/va;                      //      term to term *
trace(va*[ 1,2,3i]') ;           //
(va*[ 1,2,3i]')[1][2];           //      get coef
det ([[1,2],[-2,1]]);           //      just for matrix 1x1 et 2x2
  usefull macro to def your edp.
macro grad(u) [dx(u),dy(u)] //
macro div(u1,u2) (dx(u1)+dy(u2)) //
```


List of Plugin

```
ls /usr/local/lib/ff++/3.33/lib/
```

BEC	NewSolver	gsl	parms_FreeFem
BernadiRaugel	PARDISO	hips_FreeFem	pcm2rnm
BinaryIO	SuperLu	ilut	pipe
DxWriter	UMFPACK64	interfacepastix	ppm2rnm
Element_Mixte	VTK_writer	iovtk	qf11to25
Element_P1dc1	VTK_writer_3d	isoline	scotch
Element_P3	addNewType	isolineP1	shell
Element_P3dc	bfstream	lapack	splitedges
Element_P4	dSuperLU_DIST	lgbmo	splitmesh3
Element_P4dc	dfft	mat_dervieux	splitmesh6
Element_PkEdge	ff-lpopt	mat_psi	symmetrizeCSR
FreeFemQA	ff-NLopt	medit	tetgen
MPICG	ff-cmaes	metis	thresholdings
MUMPS	fflapack	mmg3d-v4.0	complex_SuperLU_DIST_FreeFem
MUMPS_FreeFem	ffnewuoa	mpi-cmaes	complex_pastix_FreeFem
MUMPS_seq	ffrandom	msh3	real_SuperLU_DIST_FreeFem
MetricKuate	freeyams	mshmet	real_pastix_FreeFem
MetricPk	funcTemplate	myfunction	
Morley	gmsh	myfunction2	

Important Plugin

- `qf11to25` add more quadrature formulae in 1d , 2d, and tools to build own quadrature
- `Element_*`,`Morlay`,`BernadiRaugel` add new kind of 2d finite element
- `SuperLu`,`UMFPACK64`,`SuperLu`,`MUMPS_seq` add sequential sparse solver
- `metis`,`scotch` mesh Partitioning
- `ffrandom` true random number generator : `srandomdev`,`srandom`, `random`
- `gsl` the `gsl` lib interface (lots of special function)
- `shell`,`pipe` directory and file interface, pipe interface
- `dfft` interface with `fftw3` library for FFT.
- **`msh3`**,**`tetgen`** 3d mesh tools and `tetgen` interface
- `lapack` a small `lapack`,interface of full linear solver, full eigen value problem.
- `ff-Ipopt` interface with `Ipopt` optimisation software
- `ppm2rnm` interface with `ppm` library to read `ppm` bitmap.
- `isoline` build a border from `isoline`.
- `freeyams`, `meshmet`, `mmg3d-v4`, `medit` interface of library of P. Frey to adapt mesh in 3d.

Important Plugin with MPI

- `hips_FreeFem,parms_FreeFem,MUMPS_FreeFem` parallel linear solver
- MUMPS a new version of MUMPS_FreeFem, in test.
- MPICG parallel version of CG, and GMRES
- `mpi-cmaes` parallel version of stochastic optimization algorithm.

Laplace equation, weak form

Let a domain Ω with a partition of $\partial\Omega$ in Γ_2, Γ_e .

Find u a solution in such that :

$$-\Delta u = 1 \text{ in } \Omega, \quad u = 2 \text{ on } \Gamma_2, \quad \frac{\partial u}{\partial \vec{n}} = 0 \text{ on } \Gamma_e \quad (1)$$

Denote $V_g = \{v \in H^1(\Omega) / v|_{\Gamma_2} = g\}$.

The Basic variationnal formulation with is : find $u \in V_2(\Omega)$, such that

$$\int_{\Omega} \nabla u \cdot \nabla v = \int_{\Omega} 1v + \int_{\Gamma} \frac{\partial u}{\partial n} v, \quad \forall v \in V_0(\Omega) \quad (2)$$

The finite element method is just : replace V_g with a finite element space, and the `FreeFem++` code :

Poisson equation in a fish with FreeFem++

The finite element method is just : replace V_g with a finite element space, and the FreeFem++ code :

```
mesh3 Th("fish-3d.msh");           // read a mesh 3d
fespace Vh(Th,P1);                  // define the P1 EF space

Vh u,v;                             // set test and unknown function in Vh.
macro Grad(u) [dx(u),dy(u),dz(u)]  // EOM Grad def
solve laplace(u,v,solver=CG) =
    int3d(Th) ( Grad(u)'*Grad(v) )
    - int3d(Th) ( 1*v)
    + on(2,u=2);                      // int on  $\gamma_2$ 
plot(u,fill=1,wait=1,value=0,wait=1);
```

Run:fish.edp Run:fish3d.edp

1 Introduction

2 Tools

- Remarks on weak form and boundary conditions
- Mesh generation
- Build mesh from image
- 3d mesh
- Mesh tools
- Anisotropic Mesh adaptation

3 Academic Examples

4 Numerics Tools

5 Schwarz method with overlap

Remark on `varf` versus vector or matrix

The functions appearing in the variational form are formal and local to the `varf` definition, the only important thing is the order in the parameter list, like in

```
varf vb1([u1,u2],[q]) = int2d(Th) ( (dy(u1)+dy(u2)) *q)
                        +int2d(Th) (1*q) + on(1,u1=2);
varf vb2([v1,v2],[p]) = int2d(Th) ( (dy(v1)+dy(v2)) *p)
                        +int2d(Th) (1*p);
```

To build matrix A from the bilinear part the the variational form a of type `varf` do simply

```
matrix B1 = vb1(Vh,Wh [, ...] );
matrix<complex> C1 = vb1(Vh,Wh [, ...] );
//   where the fespace have the correct number of comp.
//   Vh is "fespace" for the unknown fields with 2 comp.
//   ex fespace Vh(Th,[P2,P2]); or fespace Vh(Th,RT);
//   Wh is "fespace" for the test fields with 1 comp.
```

To build a vector, put $u1 = u2 = 0$ by setting 0 of on unknown part.

```
real[int]  b = vb2(0,Wh);
complex[int]  c = vb2(0,Wh);
```

Remark : In this case the mesh use to defined , \int , u , v can be different.

The boundary condition terms

First FreeFem++ use only the label number of edge (2d) or faces (3d).

- An "on" scalar form (for Dirichlet) : `on(1, u = g)`

The meaning is for all degree of freedom i (DoF) of this associated boundary, the diagonal term of the matrix $a_{ii} = \text{tgv}$ with the *terrible giant value* `tgv` ($=10^{30}$ by default) and the right hand side $b[i] = "(\Pi_h g)[i]" \times \text{tgv}$, where the $"(\Pi_h g)[i]"$ is the boundary DoF value given by the interpolation of g .

- An "on" vectorial form (for Dirichlet) : `on(1, u1=g1, u2=g2)` If you have vectorial finite element like RT0 , the 2 components are coupled, and so you have :
 $b[i] = "(\Pi_h(g1, g2))[i]" \times \text{tgv}$, where Π_h is the vectorial finite element interpolant.

- a linear form on Γ (for Neumann in 2d)

`-int1d(Th) (f*w)` or `-int1d(Th, 3) (f*w)`

- a bilinear form on Γ or Γ_2 (for Robin in 2d)

`int1d(Th) (K*v*w)` or `int1d(Th, 2) (K*v*w)`.

- a linear form on Γ (for Neumann in 3d)

`-int2d(Th) (f*w)` or `-int2d(Th, 3) (f*w)`

First a 10×10 grid mesh of unit square $]0,1[^2$

```
int[int] labs=[10,20,30,40];           //  bot., right, top, left
mesh Th1 = square(10,10,label=labs,region=0,[x,y]);           //
plot(Th1,wait=1);
int[int] old2newlabs=[10,11, 30,31];    //  10 -> 11, 30 -> 31
Th1=change(Th1,label=old2newlabs);      //
//  do Change in 2d or in 3d. region=a, fregion=f ,
//  flabel=f
```

a L shape domain $]0,1[^2 \setminus [\frac{1}{2},1[^2$

```
mesh Th = trunc(Th1,(x<0.5) | (y < 0.5),label=1);           //
plot(Th,cmm="Th");
mesh Thh = movemesh(Th,[-x,y]);
mesh Th3 = Th+movemesh(Th,[-x,y]);           //  glumesh ...
plot(Th3,cmm="Th3");
```

Run:mesh1.edp

a L shape domain $]0, 1[^2 \setminus [\frac{1}{2}, 1[^2$

```
border a(t=0,1.0){x=t;    y=0;    label=1;};
border b(t=0,0.5){x=1;    y=t;    label=1;};
border c(t=0,0.5){x=1-t;  y=0.5;label=1;};
border d(t=0.5,1){x=0.5;  y=t;    label=1;};
border e(t=0.5,1){x=1-t;  y=1;    label=1;};
border f(t=0.0,1){x=0;    y=1-t;label=1;};
plot(a(6) + b(4) + c(4) +d(4) + e(4) + f(6),wait=1);
mesh Th2 =buildmesh(a(6) + b(4) + c(4) +d(4) + e(4) + f(6));
```

Get a extern mesh

```
mesh Th2("april-fish.msh");
build with emc2, bamg, modulef, etc...
```

Build mesh from image 1/2

```
load "ppm2rnm" load "isoline"
real[int,int] Curves(3,1);
int[int] be(1);  int nc;                                //    nb of curve
{
  real[int,int] ff1("Lac-tyyppalanjarvi.pgm");
  int nx = ff1.n, ny=ff1.m;                             //    grey value [0,1]
  mesh Th=square(nx-1,ny-1,
                  [(nx-1)*(x), (ny-1)*(1-y)]);
  fespace Vh(Th,P1); Vh f1; f1[]=ff1;
                                     //    build border data of an isoline ...
  nc=isoline(Th,f1,iso=0.75,close=0,
             Curves,beginend=be,
             smoothing=.1,ratio=0.5);
}
```


Build mesh from image 2/2

```
border Curve0(t=0,1)                                // the extern boundary
{ int c =0;                                           // component 0
  int i0 = be[2*c], i1 = be[2*c+1]-1;
  P=Curve(xy,i0,i1,t);                               // Curve 0
  label=1;
}
...

plot(Curve1(100));                                  // show curve.
mesh Th= buildmesh(Curve1(-100));                   //
plot(Th,wait=1);                                     //
```

Run: `lac.edp`

(a lac close to Jyväskylä)



A cube with buildlayer (simple)

```
load "msh3" buildlayer
int nn=10;
int[int]
    rup=[0,2],      //    label: upper face 0-> 2 (region -> label)
    rdown=[0,1],    //    label: lower face 0-> 1 (region -> label)
    rmid=[1,1 ,2,1 ,3,1 ,4,1 ], //    4 Vert. 2d label -> 3d label
    rtet= [0,0];      //
real zmin=0,zmax=1;
mesh3 Th=buildlayers(square(nn,nn, ),nn,
    zbound=[zmin,zmax],
    region=rtet,
    labelmid=rmid,
    labelup = rup,
    labeldown = rdown);
Th= trunc(Th, ((x<0.5) | (y< 0.5) | (z<0.5)),label=3);
//    remove 1/2 cube
plot("cube",Th);
Run:Cube.edp
```

3D layer mesh of a Lac with buildlayer

```
load "msh3"//      buildlayer
load "medit"//      medit
int nn=5;
border cc(t=0,2*pi){x=cos(t);y=sin(t);label=1;}
mesh Th2= buildmesh(cc(100));
fespace Vh2(Th2,P2);
Vh2 ux,uz,p2;
int[int] rup=[0,2],  rdown=[0,1],  rmid=[1,1];
func zmin= 2-sqrt(4-(x*x+y*y));    func zmax= 2-sqrt(3.);
//      we get nn*coef layers
mesh3 Th=buildlayers(Th2,nn,
                    coef= max((zmax-zmin)/zmax,1./nn),
                    zbound=[zmin,zmax],
                    labelmid=rmid,  labelup = rup,
                    labeldown = rdown);          //      label def
medit("lac",Th);
Run:Lac.edp Run:3d-leman.edp
```

a 3d axi Mesh with buildlayer

```
func f=2*((.1+(((x/3))*(x-1)*(x-1)/1+x/100))^(1/3.)-(.1)^(1/3.));
real yf=f(1.2,0);
border up(t=1.2,0.) { x=t;y=f;label=0;}
border axe2(t=0.2,1.15) { x=t;y=0;label=0;}
border hole(t=pi,0) { x= 0.15 + 0.05*cos(t);y= 0.05*sin(t);
    label=1;}
border axel(t=0,0.1) { x=t;y=0;label=0;}
border queue(t=0,1) { x= 1.15 + 0.05*t; y = yf*t; label =0;}
int np= 100;
func bord= up(np)+axel(np/10)+hole(np/10)+axe2(8*np/10)
    + queue(np/10);
plot( bord); // plot the border ...
mesh Th2=buildmesh(bord); // the 2d mesh axi mesh
plot(Th2,wait=1);
int[int] l23=[0,0,1,1];
Th=buildlayers(Th2,coef= max(.15,y/max(f,0.05)), 50
    ,zbound=[0,2*pi],transfo=[x,y*cos(z),y*sin(z)]
    ,facemerge=1,labelmid=l23);
```

Run:3daximesh.edp

boundary mesh of a Sphere

```
load "tetgen"
mesh Th=square(10,20,[x*pi-pi/2,2*y*pi]);          //  $[-\frac{\pi}{2}, \frac{\pi}{2}] \times [0, 2\pi]$ 
func f1 =cos(x)*cos(y); func f2 =cos(x)*sin(y); func f3 = sin(x);
      // the partial derivative of the parametrization DF
func f1x=sin(x)*cos(y); func f1y=-cos(x)*sin(y);
func f2x=-sin(x)*sin(y); func f2y=cos(x)*cos(y);
func f3x=cos(x); func f3y=0;
                                     //  $M = DF^t DF$ 
func m11=f1x^2+f2x^2+f3x^2; func m21=f1x*f1y+f2x*f2y+f3x*f3y;
func m22=f1y^2+f2y^2+f3y^2;
func perio=[[4,y],[2,y],[1,x],[3,x]];
real hh=0.1/R; real vv= 1/square(hh);
Th=adaptmesh(Th,m11*vv,m21*vv,m22*vv,IsMetric=1,periodic=perio);
int[int] ref=[0,L]; // the label of the Sphere to L ( 0 -> L)
mesh3 ThS= movemesh23(Th,transfo=[f1*R,f2*R,f3*R],orientation=1,
  label=ref);
Run:Sphere.edp Run:sphere6.edp
```


Build 3d Mesh from boundary mesh

```
include "MeshSurface.idp"           //    tool for 3d surfaces meshes
mesh3 Th;
try { Th=readmesh3("Th-hex-sph.mesh"); }           //    try to read
catch(...) {           //    catch a reading error so build the mesh...
    real hs = 0.2;           //    mesh size on sphere
    int[int] NN=[11,9,10];
    real [int,int] BB=[[-1.1,1.1],[-.9,.9],[-1,1]]; //    Mesh Box
    int [int,int] LL=[[1,2],[3,4],[5,6]];           //    Label Box
    mesh3 ThHS = SurfaceHex(NN,BB,LL,1)+Sphere(0.5,hs,7,1);
                                           //    surface meshes

    real voltet=(hs^3)/6.;           //    volume mesh control.
    real[int] domaine = [0,0,0,1,voltet,0,0,0.7,2,voltet];
    Th = tetg(ThHS,switch="pqaAAYYQ",
              nbofregions=2,regionlist=domaine);
    savemesh(Th,"Th-hex-sph.mesh"); }           //    save for next run
```

- `change` to change label and region numbering in 2d and 3d.
- `movemesh` `checkmovemesh` `movemesh23` `movemesh3`
- `triangulate (2d)` , `tetgconvexhull (3d)` build mesh for a set of point
- `emptymesh (2d)` built a empty mesh for Lagrange multiplier
- `freeyams` to optimize surface mesh
- `mmg3d` to optimize volume mesh with constant surface mesh
- `mshmet` to compute metric
- `isoline` to extract isoline (2d)
- `trunc` to remove peace of mesh and split all element (2d,3d)
- `splitmesh` to split 2d mesh in no regular way.

In Euclidean geometry the length $|\gamma|$ of a curve γ of \mathbb{R}^d parametrized by $\gamma(t)_{t=0..1}$ is

$$|\gamma| = \int_0^1 \sqrt{\langle \gamma'(t), \gamma'(t) \rangle} dt$$

We introduce the metric $\mathcal{M}(x)$ as a field of $d \times d$ symmetric positive definite matrices, and the length ℓ of Γ w.r.t \mathcal{M} is :

$$\ell = \int_0^1 \sqrt{\langle \gamma'(t), \mathcal{M}(\gamma(t)) \gamma'(t) \rangle} dt$$


The key-idea is to construct a mesh where the lengths of the edges are close to 1 accordingly to \mathcal{M} .

Metrix intersection

The unit ball $\mathcal{B}\mathcal{M}$ in a metric \mathcal{M} plot the maximum mesh size on all the direction, is a ellipse.

If you we have two unknowns u and v , we just compute the metric \mathcal{M}_u and \mathcal{M}_v , find a metric \mathcal{M}_{uv} call intersection with the biggest ellipse such that :


$$\mathcal{B}(\mathcal{M}_v) \subset \mathcal{B}(\mathcal{M}_u) \cap \mathcal{B}(\mathcal{M}_v)$$



Example of mesh

$$u = (10 * x^3 + y^3) + \text{atan2}(0.001, (\sin(5 * y) - 2 * x))$$

$$v = (10 * y^3 + x^3) + \text{atan2}(0.01, (\sin(5 * x) - 2 * y)).$$




Run: `Adapt-uv.edp`

The domain is an L-shaped polygon $\Omega =]0, 1[^2 \setminus [\frac{1}{2}, 1]^2$ and the PDE is

Find $u \in H_0^1(\Omega)$ such that $-\Delta u = 1$ in Ω ,

The solution has a singularity at the reentrant angle and we wish to capture it numerically.



example of Mesh adaptation

```
int[int] lab=[1,1,1,1];
mesh Th = square(6,6,label=lab);
Th=trunc(Th,x<0.5 | y<0.5, label=1);

fespace Vh(Th,P1);          Vh u,v;          real error=0.1;
problem Problem1(u,v,solver=CG,eps=1.0e-6) =
    int2d(Th) ( dx(u)*dx(v) + dy(u)*dy(v) )
    - int2d(Th) ( v ) + on(1,u=0);
for (int i=0;i< 7;i++)
{ Problem1;                  // solving the pde
  Th=adaptmesh(Th,u,err=error,nbvx=100000);
                               // the adaptation with Hessian of u
  plot(Th,u,wait=1,fill=1);    u=u;
  error = error/ (1000.^(1./7.)); };
```

Run:CornerLap.edp

Build of the metric form the solution u

Optimal metric norm for interpolation error (function `adaptmesh` in `freefem++`) for P_1 continuous Lagrange finite element

- L^∞ : $\mathcal{M} = \frac{1}{\varepsilon} |\nabla \nabla u| = \frac{1}{\varepsilon} |\mathcal{H}|$ where $\mathcal{H} = \nabla \nabla u$
- L^p : $\mathcal{M} = \frac{1}{\varepsilon} |\det(\mathcal{H})|^{\frac{1}{2p+2}} |\mathcal{H}|$ (result of F. Alauzet, A. Dervieux)

In Norm $W^{1,p}$, the optimal metric \mathcal{M}_ℓ for the P_ℓ Lagrange finite element, Optimal is given by (with only acute triangle) (thank J-M. Mirebeau)

$$\mathcal{M}_{\ell,p} = \frac{1}{\varepsilon} (\det \mathcal{M}_\ell)^{\frac{1}{\ell p + 2}} \mathcal{M}_\ell$$

and (see `MetricPk` plugin and function)

- for P_1 : $\mathcal{M}_1 = \mathcal{H}^2$ (sub optimal with acute triangle take \mathcal{H})
- for P_2 : $\mathcal{M}_2 = 3 \sqrt{\begin{pmatrix} a & b \\ b & c \end{pmatrix}^2 + \begin{pmatrix} b & c \\ c & a \end{pmatrix}^2}$ with
 $D^{(3)}u(x, y) = (ax^3 + 3bx^2y + 3cxy^2 + dy^3)/3!$,

Run: `adapt.edp`

Run: `AdaptP3.edp`

1 Introduction

2 Tools

3 Academic Examples

- Laplace/Poisson
- 3d Poisson equation with mesh adaptation
- Linear PDE
- Linear elasticity equation
- Stokes equation
- Optimize Time depend schema

4 Numerics Tools

5 Schwarz method with overlap

Laplace equation (mixte formulation) II/III

Now we solve $-\Delta p = f$ in Ω , $p = g_d$ on Γ_d , $\partial_n p = g_n$ on Γ_n .

Γ_d, Γ_n is a partition of $\partial\Omega$.

with $\vec{u} = \nabla p$ the problem becomes :

Find \vec{u}, p such that :

$$-\nabla \cdot \vec{u} = f, \quad \vec{u} - \nabla p = 0 \quad \text{in } \Omega, \quad p = g_d \quad \text{on } \Gamma_d, \quad \partial_n p = g_n \quad \text{on } \Gamma_n \quad (3)$$

Mixte variational formulation is : find $\vec{u} \in H_{div}(\Omega)$, $p \in L^2(\Omega)$, $\vec{u} \cdot \vec{n} = g_n$ on Γ_n such that

$$\int_{\Omega} q \nabla \cdot \vec{u} + \int_{\Omega} p \nabla \cdot \vec{v} + \vec{u} \cdot \vec{v} = \int_{\Omega} -f q + \int_{\Gamma_d} g_d \vec{v} \cdot \vec{n}, \quad \forall (\vec{v}, q) \in H_{div} \times L^2, \text{ and } \vec{v} \cdot \vec{n} = 0 \text{ on } \Gamma_n$$

Laplace equation (mixed formulation) II/III

```
mesh Th=square(10,10); fespace Vh(Th,RT0), Ph(Th,P0);
func gd = 1.; func g1n = 1.; func g2n = 1.; func f = 1.;
Vh [u1,u2],[v1,v2];
Ph p,q;
solve laplaceMixte([u1,u2,p],[v1,v2,q],solver=UMFPACK)
= int2d(Th) ( p*q*0e-10 + u1*v1 + u2*v2
              + p*(dx(v1)+dy(v2)) + (dx(u1)+dy(u2))*q )
+ int2d(Th) ( f*q)
- int1d(Th,1,2,3) ( gd*(v1*N.x +v2*N.y)) // int on  $\Gamma_d$ 
+ on(4,u1=g1n,u2=g2n); // mean  $u.n = g.n$ 
```

Run:LaplaceRT.edp

Laplace equation (Garlerking discontinuous formulation) III/III

solve $-\Delta u = f$ on Ω and $u = g$ on Γ

```
macro dn(u) (N.x*dx(u)+N.y*dy(u) ) //    def the normal derivative
mesh Th = square(10,10);                //    unite square
fespace Vh(Th,P2dc);                    //    discontinuous P2 finite element
//    if pena = 0 => Vh must be P2 otherwise penalization
real pena=0;                            //    to add penalization
func f=1;    func g=0;
Vh u,v;

problem A(u,v,solver=UMFPACK) =          //
    int2d(Th) (dx(u)*dx(v)+dy(u)*dy(v) )
+   intalledges(Th) ( //    loop on all edge of all triangle
    ( jump(v)*average(dn(u)) - jump(u)*average(dn(v))
      + pena*jump(u)*jump(v) ) / nTonEdge )
-   int2d(Th) (f*v)
-   int1d(Th) (g*dn(v) + pena*g*v) ;
A; //    solve DG
```

Run:LapDG2.edp

A mathematical Poisson Problem with full Neumann BC. with 1D lagrange multiplier

The variationnall form is find $(u, \lambda) \in V_h \times \mathbb{R}$ such that

$$\forall (v, \mu) \in V_h \times \mathbb{R} \quad a(u, v) + b(u, \mu) + b(v, \lambda) = l(v), \quad \text{where } b(u, \mu) = \mu \int_{\Omega} u$$

```
mesh Th=square(10,10);      fespace Vh(Th,P1);      //      P1 FE space
int n = Vh.ndof,  n1 = n+1; func f=1+x-y;
macro Grad(u) [dx(u),dy(u)]                                     //      EOM
varf va(uh,vh) = int2d(Th) ( Grad(uh)'*Grad(vh)  );
varf vL(uh,vh) = int2d(Th) ( f*vh  );
varf vb(uh,vh)=  int2d(Th) (1.*vh);
matrix A=va(Vh,Vh);
real[int] b=vL(0,Vh), B = vb(0,Vh);
real[int] bb(n1),x(n1),b1(1),l(1); b1=0;
matrix AA = [ [ A ,  B ] , [ B', 0 ] ];  bb = [ b, b1];
set(AA,solver=UMFPACK);      //      set the type of linear solver.
x = AA^-1*bb;      [uh[],l] = x;      //      solve the linear systeme
plot(uh,wait=1);      //      set the value
```

Run:Laplace-lagrange-mult.edp

Poisson equation with 3d mesh adaptation

```
load "msh3" load "tetgen" load "mshmet" load "medit"
int nn = 6;
int[int] l1111=[1,1,1,1],l01=[0,1],l11=[1,1]; // label numbering
mesh3 Th3=buildlayers(square(nn,nn,region=0,label=l1111),
    nn, zbound=[0,1], labelmid=l11,labelup = l01,labeldown = l01);
Th3=trunc(Th3,(x<0.5)|(y < 0.5)|(z < 0.5) ,label=1); // remove ]0.5,1[3

fespace Vh(Th3,P1); Vh u,v; // FE. space definition
macro Grad(u) [dx(u),dy(u),dz(u)] // EOM
problem Poisson(u,v,solver=CG) =
    int3d(Th3) ( Grad(u)'*Grad(v) ) -int3d(Th3) ( 1*v ) + on(1,u=0);

real errm=1e-2; // level of error
for(int ii=0; ii<5; ii++)
{
    Poisson; Vh h;
    h[]=mshmet(Th3,u,normalization=1,aniso=0,nbregul=1,hmin=1e-3,
        hmax=0.3,err=errm);
    errm*= 0.8; // change the level of error
    Th3=tetgreconstruction(Th3,switch="raAQ"
        ,sizeofvolume=h*h*h/6.);
    medit("U-adap-iso-"+ii,Th3,u,wait=1);
}
```

Run:Laplace-Adapt-3d.edp

Linear Lamé equation, weak form

Let a domain $\Omega \subset \mathbb{R}^d$ with a partition of $\partial\Omega$ in Γ_d, Γ_n .

Find the displacement \mathbf{u} field such that :

$$-\nabla \cdot \sigma(\mathbf{u}) = \mathbf{f} \text{ in } \Omega, \quad \mathbf{u} = \mathbf{0} \text{ on } \Gamma_d, \quad \sigma(\mathbf{u}) \cdot \mathbf{n} = \mathbf{0} \text{ on } \Gamma_n \quad (4)$$

Where $\varepsilon(\mathbf{u}) = \frac{1}{2}(\nabla \mathbf{u} + {}^t\nabla \mathbf{u})$ and $\sigma(\mathbf{u}) = \mathbf{A}\varepsilon(\mathbf{u})$ with \mathbf{A} the linear positif operator on symmetric $d \times d$ matrix corresponding to the material propriety. Denote

$$V_{\mathbf{g}} = \{\mathbf{v} \in H^1(\Omega)^d / \mathbf{v}|_{\Gamma_d} = \mathbf{g}\}.$$

The Basic displacement variational formulation is : find $\mathbf{u} \in V_0(\Omega)$, such that :

$$\int_{\Omega} \varepsilon(\mathbf{v}) : \mathbf{A}\varepsilon(\mathbf{u}) = \int_{\Omega} \mathbf{v} \cdot \mathbf{f} + \int_{\Gamma} ((\mathbf{A}\varepsilon(\mathbf{u})) \cdot \mathbf{n}) \cdot \mathbf{v}, \quad \forall \mathbf{v} \in V_0(\Omega) \quad (5)$$

Linear elasticity equation, in FreeFem++

The finite element method is just : replace V_g with a finite element space, and the FreeFem++ code :

```
load "medit"    include "cube.idp"
int[int]  Nxyz=[20,5,5];
real [int,int]  Bxyz=[[0.,5.],[0.,1.],[0.,1.]];
int [int,int]  Lxyz=[[1,2],[2,2],[2,2]];
mesh3 Th=Cube (Nxyz,Bxyz,Lxyz);

//      Alu ...
real rhoAlu = 2600, alu11= 1.11e11 , alu12 = 0.61e11;
real alu44= (alu11-alu12)*0.5;
func Aalu = [  [alu11, alu12,alu12,    0.    ,0.    ,0.    ],
               [alu12, alu11,alu12,    0.    ,0.    ,0.    ],
               [alu12, alu12,alu11,    0.    ,0.    ,0.    ],
               [0.    , 0.    , 0.    , alu44,0.    ,0.    ],
               [0.    , 0.    , 0.    , 0.    ,alu44,0.    ],
               [0.    , 0.    , 0.    , 0.    ,0.    ,alu44]  ];

real gravity = -9.81;
```



Linear elasticity equation, in FreeFem++

```
fespace Vh(Th, [P1,P1,P1]);
Vh [u1,u2,u3], [v1,v2,v3];
macro Strain(u1,u2,u3)
  [ dx(u1), dy(u2), dz(u3),
    (dz(u2) +dy(u3)), (dz(u1)+dx(u3)),
    (dy(u1)+dx(u2)) ]
solve Lamé([u1,u2,u3], [v1,v2,v3])=
  int3d(Th) (
    Strain(v1,v2,v3)'*(Aalu*Strain(u1,u2,u3))
- int3d(Th) ( rhoAlu*gravity*v3)
  + on(1,u1=0,u2=0,u3=0) ;


real coef= 0.1/u1[]*linfty;  int[int] ref2=[1,0,2,0];
mesh3 Thm=movemesh3(Th,
  transfo=[x+u1*coef,y+u2*coef,z+u3*coef],
  label=ref2);
plot(Th,Thm, wait=1,cmm="coef amplification = "+coef );
medit ("Th-Thm",Th,Thm);
```

// EOM

cond amplification = 3997.95



Run:beam-3d.edp



Run:beam-EV-3d.edp

Run:beam-3d-Adapt.edp

The Stokes equation is find a velocity field $\mathbf{u} = (u_1, \dots, u_d)$ and the pressure p on domain Ω of \mathbb{R}^d , such that

$$\begin{aligned} -\Delta \mathbf{u} + \nabla p &= 0 && \text{in } \Omega \\ \nabla \cdot \mathbf{u} &= 0 && \text{in } \Omega \\ \mathbf{u} &= \mathbf{u}_\Gamma && \text{on } \Gamma \end{aligned}$$

where \mathbf{u}_Γ is a given velocity on boundary Γ .

The classical variational formulation is : Find $\mathbf{u} \in H^1(\Omega)^d$ with $\mathbf{u}|_\Gamma = \mathbf{u}_\Gamma$, and $p \in L^2(\Omega)/\mathbb{R}$ such that

$$\forall \mathbf{v} \in H_0^1(\Omega)^d, \forall q \in L^2(\Omega)/\mathbb{R}, \quad \int_{\Omega} \nabla \mathbf{u} : \nabla \mathbf{v} - p \nabla \cdot \mathbf{v} - q \nabla \cdot \mathbf{u} = 0$$

or now find $p \in L^2(\Omega)$ such than (with $\varepsilon = 10^{-10}$)

$$\forall \mathbf{v} \in H_0^1(\Omega)^d, \forall q \in L^2(\Omega), \quad \int_{\Omega} \nabla \mathbf{u} : \nabla \mathbf{v} - p \nabla \cdot \mathbf{v} - q \nabla \cdot \mathbf{u} + \varepsilon p q = 0$$

```
... build mesh .... Th (3d) T2d ( 2d)
fespace VVh(Th, [P2,P2,P2,P1]);           // Taylor Hood FE.
macro Grad(u) [dx(u),dy(u),dz(u)]        // EOM
macro div(u1,u2,u3) (dx(u1)+dy(u2)+dz(u3)) // EOM
VVh [u1,u2,u3,p], [v1,v2,v3,q];
solve vStokes([u1,u2,u3,p],[v1,v2,v3,q]) =
  int3d(Th) (
    Grad(u1)'*Grad(v1)
    + Grad(u2)'*Grad(v2)
    + Grad(u3)'*Grad(v3)
    - div(u1,u2,u3)*q - div(v1,v2,v3)*p
    - 1e-10*q*p )
+ on(1,u1=0,u2=0,u3=0) + on(2,u1=1,u2=0,u3=0);
```

Run:Stokes3d.edp

Fast method for Time depend Problem / formulation

First, it is possible to define variational forms, and use this forms to build matrix and vector to make very fast script (4 times faster here).

For example solve the Thermal Conduction problem of section 3.4. We must solve the temperature equation in Ω in a time interval $(0,T)$.

$$\begin{aligned} \partial_t u - \nabla \cdot (\kappa \nabla u) &= 0 \text{ in } \Omega \times (0, T), \\ u(x, y, 0) &= u_0 + xu_1 \\ u &= 30 \text{ on } \Gamma_{24} \times (0, T), \quad \kappa \frac{\partial u}{\partial n} + \alpha(u - u_e) = 0 \text{ on } \Gamma \times (0, T). \end{aligned} \quad (6)$$

The variational formulation is in $L^2(0, T; H^1(\Omega))$; we shall seek u^n satisfying

$$\forall w \in V_0; \quad \int_{\Omega} \frac{u^n - u^{n-1}}{\delta t} w + \kappa \nabla u^n \nabla w + \int_{\Gamma} \alpha(u^n - u_{ue}) w = 0$$

where $V_0 = \{w \in H^1(\Omega) / w|_{\Gamma_{24}} = 0\}$.

Fast method for Time depend Problem algorithm

So the to code the method with the matrices $A = (A_{ij})$, $M = (M_{ij})$, and the vectors $u^n, b^n, b', b'', b_{cl}$ (notation if w is a vector then w_i is a component of the vector).

$$u^n = A^{-1}b^n, \quad b' = b_0 + Mu^{n-1}, \quad b'' = \frac{1}{\varepsilon} b_{cl}, \quad b_i^n = \begin{cases} b''_i & \text{if } i \in \Gamma_{24} \\ b'_i & \text{else} \end{cases}$$

Where with $\frac{1}{\varepsilon} = \text{tgv} = 10^{30}$:

$$\begin{aligned} A_{ij} &= \begin{cases} \int_{\Omega} w_j w_i / dt + k(\nabla w_j \cdot \nabla w_i) + \int_{\Gamma_{13}} \alpha w_j w_i & \text{if } i \in \Gamma_{24}, \text{ and } j = i \\ \int_{\Omega} w_j w_i / dt & \text{else} \end{cases} \\ M_{ij} &= \begin{cases} \int_{\Omega} w_j w_i / dt & \text{if } i \in \Gamma_{24}, \text{ and } j = i \\ \int_{\Omega} w_j w_i / dt & \text{else} \end{cases} \\ b_{0,i} &= \int_{\Gamma_{13}} \alpha u_{ue} w_i \\ b_{cl} &= u^0 \quad \text{the initial data} \end{aligned}$$

```
...  
Vh u0=fu0,u=u0;  
Create three variational formulation, and build the matrices  $A, M$ .  
varf vthermic (u,v)= int2d(Th) (u*v/dt  
                + k*(dx(u) * dx(v) + dy(u) * dy(v)))  
  + int1d(Th,1,3) (alpha*u*v)  + on(2,4,u=1);  
varf vthermic0(u,v) = int1d(Th,1,3) (alpha*ue*v);  
varf vMass (u,v)= int2d(Th) ( u*v/dt)  + on(2,4,u=1);  
  
real tgv = 1e30;  
matrix A= vthermic(Vh,Vh,tgv=tgv,solver=CG);  
matrix M= vMass(Vh,Vh);
```

Fast The Time depend Problem/ edp

Now, to build the right hand size we need 4 vectors.

```
real[int]  b0 = vthermic(0,Vh);    //    constant part of RHS
real[int]  bcn = vthermic(0,Vh);    //    tgv on Dirichlet part
          //    we have for the node  $i : i \in \Gamma_{24} \Leftrightarrow bcn[i] \neq 0$ 
real[int]  bcl=tgv*u0[];           //    the Dirichlet B.C. part
```

The Fast algorithm :

```
for(real t=0;t<T;t+=dt){
  real[int] b = b0;                //    for the RHS
  b += M*u[];                      //    add the the time dependent part
  b = bcn? bcl : b; //    do  $\forall i: b[i] = bcn[i]? bcl[i] : b[i];$ 
  u[] = A^-1*b;                    //    Solve linear problem
  plot(u);
}
```

Run:Heat.edp

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools**
 - Connectivity
 - Input/Output
 - Eigenvalue
 - Eigenvalue/ Eigenvector
 - Optimization Tools
 - MPI/Parallel
- 5 Schwarz method with overlap

```
mesh Th=square(5,5);
fespace Wh(Th,P2);
cout << " nb of DoF      : " << Wh.ndof << endl;
cout << " nb of DoF / K : " << Wh.ndofK << endl;
int k= 2, kdf= Wh.ndofK;;                                //      element 2
cout << " df of element " << k << ":";
for (int i=0;i<kdf;i++)  cout << Wh(k,i) << " ";
cout << endl;
```

Remark on local numbering of Dof by element is
for each sub finite element P_k in $[P_2, P_2, P_1]$ get first DoF on vertex, second DoF on
edge (opposite to vertex), second on K.

Run:Mesh-info.edp

uses `cout`, `cin`, `endl`, `<<`, `>>`.

To write to (resp. read from) a file,

declare a new variable `ofstream ofile("filename");`

or

`ofstream ofile("filename", append); (resp. ifstream`

`ifile("filename");)`

and use `ofile (resp. ifile)` as `cout (resp. cin)`.

You can use pipe to transfer data to a other code here (gnuplot), see `pipe.edp` example :

Run:pipe.edp

Run:io.edp

The problem, Find the first λ, u_λ such that :

$$a(u_\lambda, v) = \int_{\Omega} \nabla u_\lambda \nabla v = \lambda \int_{\Omega} u_\lambda v = \lambda b(u_\lambda, v)$$

the boundary condition is make with exact penalization : we put $1e30 = tgv$ on the diagonal term of the lock degree of freedom. So take Dirichlet boundary condition only with a variational form and not on b variational form , because we compute eigenvalue of

$$\frac{1}{\lambda} v = A^{-1} B v$$

Otherwise we get spurious mode.

Arpack interface :

```
int k=EigenValue (A,B,sym=true,value=ev,vector=eV) ;
```

Eigenvalue/ Eigenvector example code

```
...
fespace Vh(Th,P1);
macro Grad(u) [dx(u),dy(u),dz(u)] // EOM
varf a(u1,u2)= int3d(Th)( Grad(u1)'*Grad(u2) + on(1,u1=0) ;
varf b([u1],[u2]) = int3d(Th)( u1*u2 ) ; // no BC
matrix A= a(Vh,Vh,solver=UMFPACK),
        B= b(Vh,Vh,solver=CG,eps=1e-20);

int nev=40; // number of computed eigenvalue close to 0
real[int] ev(nev); // to store nev eigenvalue
Vh[int] eV(nev); // to store nev eigenvector
int k=EigenValue(A,B,sym=true,value=ev,vector=eV);
k=min(k,nev);
for (int i=0;i<k;i++)
    plot(eV[i],cmm="ev "+i+" v =" + ev[i],wait=1,value=1);
Execute Lap3dEigenValue.edp Execute LapEigenValue.edp
```

The IPOPT optimizer in a FreeFem++ script is done with the `IPOPT` function included in the `ff-Ipopt` dynamic library. IPOPT is designed to solve constrained minimization problem in the form :

$$\begin{array}{ll} \text{find} & x_0 = \underset{x \in \mathbb{R}^n}{\operatorname{argmin}} f(x) \\ \text{s.t.} & \left\{ \begin{array}{ll} \forall i \leq n, & x_i^{\text{lb}} \leq x_i \leq x_i^{\text{ub}} & (\text{simple bounds}) \\ \forall i \leq m, & c_i^{\text{lb}} \leq c_i(x) \leq c_i^{\text{ub}} & (\text{constraints functions}) \end{array} \right. \end{array}$$

Where `ub` and `lb` stand for "upper bound" and "lower bound". If for some $i, 1 \leq i \leq m$ we have $c_i^{\text{lb}} = c_i^{\text{ub}}$, it means that c_i is an equality constraint, and an inequality one if $c_i^{\text{lb}} < c_i^{\text{ub}}$.

```
func real J(real[int] &X) {...}           // Fitness Function,
func real[int] gradJ(real[int] &X) {...}   // Gradient

func real[int] C(real[int] &X) {...}       // Constraints
func matrix jacC(real[int] &X) {...}      // Constraints jacobian

matrix jacCBuffer;                        // just declare, no need to define yet
func matrix jacC(real[int] &X)
{
    ...                                   // fill jacCBuffer
    return jacCBuffer;
}
```

The hessian returning function is somewhat different because it has to be the hessian of the lagrangian

function : $(x, \sigma_f, \lambda) \mapsto \sigma_f \nabla^2 f(x) + \sum_{i=1}^m \lambda_i \nabla^2 c_i(x)$ where $\lambda \in \mathbb{R}^m$ and $\sigma \in \mathbb{R}$. Your hessian function should

then have the following prototype :

```
matrix hessianLBuffer;                    // just to keep it in mind
func matrix hessianL(real[int] &X, real sigma, real[int] &lambda) {...}
```

```
real[int] Xi = ...;                                // starting point
IPOPT(J,gradJ,hessianL,C,jacC,Xi, ... );

IPOPT(J,gradJ,C,jacC,Xi,...);                       // IPOPT with BFGS
IPOPT(J,gradJ,hessianJ,Xi,...);                     // Newton IPOPT
                                                    // without constraints
IPOPT(J,gradJ,Xi, ... );                             // BFGS, no constraints
IPOPT(J,gradJ,Xi, ... );                             // BFGS, no constraints
IPOPT([b,A],CC,uil[],lb=lb1[],clb=cl[]..);           // affine case
...
```



```
load "ff-Ipopt"
varf vP([u1,u2],[v1,v2]) = int2d(Th) (Grad(u1)'*Grad(v1)+ Grad(u2)'*Grad(v2))
- int2d(Th) (f1*v1+f2*v2);

matrix A = vP(Vh,Vh); // Fitness function matrix...
real[int] b = vP(0,Vh); // and linear form
int[int] II1=[0],II2=[1]; // Constraints matrix
matrix C1 = interpolate (Wh,Vh, U2Vc=II1);
matrix C2 = interpolate (Wh,Vh, U2Vc=II2);
matrix CC = -1*C1 + C2; // u2 - u1 > 0
Wh cl=0; // constraints lower bounds (no upper bounds)
varf vGamma([u1,u2],[v1,v2]) = on(1,2,3,4,u1=1,u2=1);
real[int] onGamma=vGamma(0,Vh);
Vh [ub1,ub2]=[g1,g2];
Vh [lb1,lb2]=[g1,g2];
ub1[] = onGamma? ub1[] : 1e19 ; // Unbounded in interior
lb1[] = onGamma? lb1[] : -1e19 ;
Vh [uzi,uzi2]=[uz,uz2],[lzi,lzi2]=[lz,lz2],[ui1,ui2]=[u1,u2];
Wh lmi=lm;
IPOPT([b,A],CC,ui1[],lb=lb1[],clb=cl[],ub=ub1[],warmstart=iter>1,uz=uzi[],lz=lzi[],lm=lmi)
```

Run:IpoptLap.edp

Run:IpoptVI2.edp

Run:IpoptMinSurfVol.edp

```
load "ff-NLopt"
...
if(kas==1)
  mini = nloptAUGLAG(J, start, grad=dJ, lb=lo, ub=up, IConst=IneqC,
    gradIConst=dIneqC, subOpt="LBFGS", stopMaxFEval=10000,
    stopAbsFTol=starttol);
else if(kas==2)
  mini = nloptMMA(J, start, grad=dJ, lb=lo, ub=up, stopMaxFEval=10000,
    stopAbsFTol=starttol);
else if(kas==3)
  mini = nloptAUGLAG(J, start, grad=dJ, IConst=IneqC, gradIConst=dIneqC,
    EConst=BC, gradEConst=dBC,
    subOpt="LBFGS", stopMaxFEval=200, stopRelXTol=1e-2);
else if(kas==4)
  mini = nloptSLSQP(J, start, grad=dJ, IConst=IneqC, gradIConst=dIneqC,
    EConst=BC, gradEConst=dBC,
    stopMaxFEval=10000, stopAbsFTol=starttol);
```

Run:VarIneq2.edp

Stochastic interface

This algorithm works with a normal multivariate distribution in the parameters space and try to adapt its covariance matrix using the information provides by the successive function evaluations. Syntax : `cmaes(J,u[],...)` ()

From <http://www.lri.fr/~hansen/javadoc/fr/inria/optimization/cmaes/package-summary.html>

Stochastic Exemple

```
load "ff-cmaes"

real mini = cmaes(J,start,stopMaxFunEval=10000*(al+1),
                 stopTolX=1.e-4/(10*(al+1)),
                 initialStdDev=(0.025/(pow(100.,al))));
SSPToFEF(best1[],best2[],start);
```

Run:cmaes-VarIneq.edp

```
load "mpi-cmaes"

real mini = cmaesMPI(J,start,stopMaxFunEval=10000*(al+1),
                    stopTolX=1.e-4/(10*(al+1)),
                    initialStdDev=(0.025/(pow(100.,al))));
SSPToFEF(best1[],best2[],start);
```

remark, the FreeFem `mpicommworld` is used by default. The user can specify his own MPI communicator with the named parameter "`comm=`", see the MPI section of this manual for more informations about communicators in FreeFem++.

A first way to break complexity

- 1 Build matrix in parallel by assembling par region remark with the change function you change the region numbering to build region.

```
real c = mpisize/real(Th.nt) ;  
Th=change(Th,fregion= min(mpisize-1,int(nuTriangle*c))) ;
```

- 2 Assemble the full matrix

```
varf vlaplace(uh,vh) = // definition de problem  
    int3d(Th,mpirank) ( uh*vh+ dt*Grad(uh)'*grad(vh) )  
    + int3d(Th,mpirank) ( dt*vh*f) + on(1,uh=g) ;  
matrix A,Ai = vlaplace(Vh,Vh,tgv=ttgv) ;  
mpiAllReduce(Ai,A,mpiCommWorld,mpiSUM) ; // assemble in //
```

- 3 Solve the linear using a good parallel solver (MUMPS)

```
load "MUMPS_FreeFem"  
uh[] = A^-1*b ; // resolution
```

Run:Heat3d.edp

Run:NSCaraCyl-100-mpi.edp

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools
- 5 Schwarz method with overlap**
 - Poisson equation with Schwarz method
 - Transfer Part
 - parallel GMRES
 - A simple Coarse grid solver
 - Numerical experiment

- 6 Advance exercice

To solve the following Poisson problem on domain Ω with boundary Γ in $L^2(\Omega)$:

$$-\Delta u = f, \text{ in } \Omega, \text{ and } u = g \text{ on } \Gamma,$$

where $f \in L^2(\Omega)$ and $g \in H^{\frac{1}{2}}(\Gamma)$ are two given functions.

Let introduce $(\pi_i)_{i=1,\dots,N_p}$ a positive regular partition of the unity of Ω , q-e-d :

$$\pi_i \in \mathcal{C}^0(\Omega) : \quad \pi_i \geq 0 \text{ and } \sum_{i=1}^{N_p} \pi_i = 1.$$

Denote Ω_i the sub domain which is the support of π_i function and also denote Γ_i the boundary of Ω_i .

The parallel Schwarz method is Let $\ell = 0$ the iterator and a initial guest u^0 respecting the boundary condition (i.e. $u^0|_{\Gamma} = g$).

$$\forall i = 1.., N_p : \quad -\Delta u_i^\ell = f, \text{ in } \Omega_i, \quad \text{and } u_i^\ell = u^\ell \text{ on } \Gamma_i \quad (7)$$

$$u^{\ell+1} = \sum_{i=1}^{N_p} \pi_i u_i^\ell \quad (8)$$

Some Remark

We never use finite element space associated to the full domain Ω because it is too expensive. So we use on each domain i we defined $J_i = \{j \in 1, \dots, N_p / \Omega_i \cap \Omega_j \neq \emptyset\}$ and we have

$$(u^{\ell+1})|_{\Omega_i} = \sum_{j \in J_i} (\pi_j u_j^\ell)|_{\Omega_i} \quad (9)$$

We denote $u_{h|i}^\ell$ the restriction of u_h^ℓ on V_{hi} , so the discrete problem on Ω_i of problem (7) is find $u_{hi}^\ell \in V_{hi}$ such that :

$$\forall v_{hi} \in V_{0i} : \int_{\Omega_i} \nabla v_{hi} \cdot \nabla u_{hi}^\ell = \int_{\Omega_i} f v_{hi},$$

$$\forall k \in \mathcal{N}_{hi}^{\Gamma_i} : \sigma_i^k(u_{hi}^\ell) = \sigma_i^k(u_{h|i}^\ell)$$

where $\mathcal{N}_{hi}^{\Gamma_i}$ is the set of the degree of freedom (Dof) on $\partial\Omega_i$ and σ_i^k the Dof of V_{hi} .

Transfer Part equation(5)

To compute $v_i = (\pi_i u_i)|_{\Omega_i} + \sum_{j \in J_i} (\pi_j u_j)|_{\Omega_i}$ and can be write the freefem++ function Update with asynchronous send/recv (**Otherwise dead lock**).

```
func bool Update(real[int] &ui, real[int] &vi)
{
  int n= jpart.n;
  for(int j=0;j<njpart;++j)  Usend[j][]=sMj[j]*ui;
  mpiRequest[int]  rq(n*2);
  for (int j=0;j<n;++j)
    Irecv(processor(jpart[j],comm,rq[j  ]), Ri[j][]);
  for (int j=0;j<n;++j)
    Isend(processor(jpart[j],comm,rq[j+n]), Si[j][]);
  for (int j=0;j<n*2;++j)
    int k= mpiWaitAny(rq);
  vi = Pii*ui;
                                     //      set to  $(\pi_i u_i)|_{\Omega_i}$ 
                                     //      apply the unity local partition .
  for(int j=0;j<njpart;++j)
    vi += rMj[j]*Vrecv[j][];
                                     //      add  $(\pi_j u_j)|_{\Omega_i}$ 
  return true; }

```

Finally you can easily accelerate the fixe point algorithm by using a parallel GMRES algorithm after the introduction the following affine \mathcal{S}_i operator sub domain Ω_i .

```
func real[int] Si(real[int]& U) {
  real[int] V(U.n) ; b= onG .* U;
  b = onG? b : Bi;
  V = Ai^-1*b; // (7)
  Update(V,U); // (??)
  V -= U; return V; }
```

Where the parallel MPIGMRES or MPICG algorithm is to solve $A_i x_i = b_i, i = 1, \dots, N_p$ by just changing the dot product by reduce the local dot product of all process with the following MPI code :

```
template<class R> R ReduceSum1(R s,MPI_Comm * comm)
{
  R r=0;
  MPI_Allreduce( &s, &r, 1 ,MPI_TYPE<R>::TYPE(),
                MPI_SUM, *comm );
  return r; }
```

A simple coarse grid is we solve the problem on the coarse grid :



```
func bool CoarseSolve(real[int]& V, real[int]& U,
                    mpiComm& comm)
{
    if(AC.n==0 && mpiRank(comm)==0)           // first time build
        AC = vPbC(VhC, VhC, solver=sparsesolver);
    real[int] Uc(Rci.n), Bc(Uc.n);
    Uc= Rci*U;                                // Fine to Coarse
    mpiReduce(Uc, Bc, processor(0, comm), mpiSUM);
    if(mpiRank(comm)==0)
        Uc = AC^-1*Bc;                        // solve of proc 0
    broadcast(processor(0, comm), Uc);
    V = Pci*Uc;                                // Coarse to Fine
}
```

Limitation : if the initial problem, data have oscillation, you must use homogenization technic on coarse problem, or use the F. Nataf and co, preconditionner.

So we finally we get 4 algorithms

- 1 The basic schwarz algorithm $u^{\ell+1} = \mathcal{S}(u^\ell)$, where \mathcal{S} is one iteration of schwarz process.
- 2 Use the GMRES to find u solution of the linear system $\mathcal{S}u - u = 0$.
- 3 Use the GMRES to solve parallel problem $\mathcal{A}_i u_i = b_i$, $i = 1, \dots, N_p$, with RAS preconditionneur
- 4 Use the method with two level preconditionneur RAS and Coarse.

On the SGI UV 100 of the lab :



A Parallel Numerical experiment on laptop

We consider first example in an academic situation to solve Poisson Problem on the cube $\Omega =]0, 1[^3$

$$-\Delta u = 1, \text{ in } \Omega; \quad u = 0, \text{ on } \partial\Omega. \quad (10)$$

With a cartesian meshes \mathcal{T}_{hn} of Ω with $6n^3$ tetrahedron, the coarse mesh is \mathcal{T}_{hm}^* , and m is a divisor of n .

We do the validation of the algorithm on a Laptop Intel Core i7 with 4 core at 1.8 Ghz with 4Go of RAM DDR3 at 1067 Mhz,

Run:DDM-Schwarz-Lap-2dd.edp

Run:DDM-Schwarz-Lame-3d.edp

Run:DDM-Schwarz-Lame-2d.edp

Run:DDM-Schwarz-Stokes-2d.edp

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools
- 5 Schwarz method with overlap
- 6 Advance exercice**
 - Heat equation with thermic resistance
- 7 No Linear Problem

Heat equation with thermic resistance

let Ω be a domain cut with internal boundary Γ_0 in 2 sub-domain $\Omega_i, (i = 1, 2)$
We have Heat equation (Poisson) on Ω , but on Γ_0 we have a jump $[u]$ on the temperature u proportional to the temperature flux which is continue

So the equation to solve is :

Find u such that $u|_{\Omega_i} \in H(\Omega_i)$ for $i = 1, 2$ and

$$-\nabla \kappa \nabla u = f_i, \quad \text{in } \Omega_i$$

$$\alpha[u] - \kappa \nabla u \cdot n = 0, \quad [\kappa \nabla u \cdot n] = 0, \quad \text{on } \Gamma_0$$

+ external boundary condition on $\partial\Omega$.

For the test take :

$L = 3, \Omega =]-L, L[\times]0, 1[, \Gamma_0 = \{\sin(\pi y)/5, y \in [0, 1]\}$, take $\kappa = i$ in Ω_i .

The external boundary condition on $\partial\Omega$ are : $\kappa \nabla u \cdot n = 0$ on upper and lower boundary
, $u = 0$ at the left part, $u = 1$ at the right part.

Heat equation with thermic resistance

Method 1 : Solve 2 coupled problems and use the block matrix tools to defined the linear system of the problem.

Method 2 : We suppose the Γ_0 move with time, and Γ_0 is not discretize in the mesh.

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools
- 5 Schwarz method with overlap
- 6 Advance exercice
- 7 No Linear Problem**
 - Navier-Stokes
 - Variational Inequality
 - Ground water

To solve $F(u) = 0$ the Newton's algorithm is

- 1 u^0 a initial guest
- 2 do
 - find w^n solution of $DF(u^n)w^n = F(u^n)$
 - $u^{n+1} = u^n - w^n$
 - if($\|w^n\| < \varepsilon$) break ;

For Navier Stokes problem the algorithm is : $\forall v, q,$

$$F(u, p) = \int_{\Omega} (u \cdot \nabla) u \cdot v + u \cdot v + \nu \nabla u : \nabla v - q \nabla \cdot u - p \nabla \cdot v + BC$$

$$\begin{aligned} DF(u, p)(w, w_p) &= \int_{\Omega} (w \cdot \nabla) u \cdot v + (u \cdot \nabla) w \cdot v \\ &\quad + \int_{\Omega} \nu \nabla w : \nabla v - q \nabla \cdot w - w_p \nabla \cdot v + BC0 \end{aligned}$$

Run:cavityNewton.edp

Run:NSNewtonCyl-100-mpi.edp

$$\frac{\partial u}{\partial t} + (u \cdot \nabla)u - \nu \Delta u + \nabla p = 0, \quad \nabla \cdot u = 0$$

with the same boundary conditions and with initial conditions $u = 0$.

This is implemented by using the interpolation operator for the term $\frac{\partial u}{\partial t} + (u \cdot \nabla)u$, giving a discretization in time

$$\begin{aligned} \frac{1}{\tau}(u^{n+1} - u^n \circ X^n) - \nu \Delta u^{n+1} + \nabla p^{n+1} &= 0, \\ \nabla \cdot u^{n+1} &= 0 \end{aligned} \tag{11}$$

The term $X^n(x) \approx x - \tau u^n(x)$ will be computed by the interpolation operator or convect operator.

Or better we use an order 2 schema, BDF1

$$\frac{\partial u}{\partial t} + (u \cdot \nabla)u \approx \frac{(3u^{n+1} - 4u^n \circ X_1^n + u^{n-1} \circ X_2^n)}{2\tau}$$

with $u^* = 2u^n - u^{n-1}$, and $X_1^n(x) \approx x - \tau u^*(x)$, $X_2^n(x) \approx x - 2\tau u^*(x)$

Run: NSCaraCyl-100-mpi.edp

The ff++ NSI 3d code

```
real alpha =1./dt;
varf vNS([uu1,uu2,uu3,p],[v1,v2,v3,q]) =
  int3d(Th) ( alpha*(uu1*v1+uu2*v2+uu3*v3)
    + nu*(Grad(uu1)'*Grad(v1)+Grad(uu2)'*Grad(v2)
+Grad(uu3)'*Grad(v3))
    - div(uu1,uu2,uu3)*q - div(v1,v2,v3)*p + 1e-10*q*p )
    + on(1,2,3,4,5,uu1=0,uu2=0,uu3=0)
    + on(6,uu1=4*(1-x)*(x)*(y)*(1-y),uu2=0,uu3=0)
    + int3d(Th) ( alpha*(
      u1(X1,X2,X3)*v1 + u2(X1,X2,X3)*v2 + u3(X1,X2,X3)*v3 ));
A = vNS(VVh,VVh); set(A,solver=UMFPACK); // build and factorize
matrix
real t=0;
for(int i=0;i<50;++i)
{ t += dt; X1[]=XYZ[]-u1[]*dt; // set  $\chi=[X1,X2,X3]$  vector
  b=vNS(0,VVh); // build NS rhs
  u1[]= A^-1 * b; // solve the linear systeme
  ux= u1(x,0.5,y); uz= u3(x,0.5,y); p2= p(x,0.5,y);
  plot([ux,uz],p2,cmm=" cut y = 0.5, time =" +t,wait=0); }
```

Variational Inequality

To solve just make a change of variable $u = u^+ - u^-$, $u > 0$ and $v = u^+ + u^-$, and we get a classical VI problem on u and the Poisson on v .

So we can use the algorithm of Primal-Dual Active set strategy as a semi smooth Newton Method HinterMuller, K. Ito, K. Kunisch SIAM J. Optim. V 13, I 3, 2002.

In this case, we just do all implementation by hand in FreeFem++ language


Run: VI-2-membrane-adap.edp

A Free Boundary problem , (phreatic water)

Let a trapezoidal domain Ω defined in FreeFem++ :

```
real L=10; // Width
real h=2.1; // Left height
real h1=0.35; // Right height
border a(t=0,L){x=t;y=0;label=1;}; // impermeable  $\Gamma_a$ 
border b(t=0,h1){x=L;y=t;label=2;}; // the source  $\Gamma_b$ 
border f(t=L,0){x=t;y=t*(h1-h)/L+h;label=3;}; //  $\Gamma_f$ 
border d(t=h,0){x=0;y=t;label=4;}; // Left impermeable  $\Gamma_d$ 
int n=10;
mesh Th=buildmesh (a(L*n)+b(h1*n)+f(sqrt(L^2+(h-h1)^2)*n)+d(h*n));
plot(Th,ps="dTh.eps");
```

The initial mesh



The problem is : find p and Ω such that :

$$\left\{ \begin{array}{ll} -\Delta p = 0 & \text{in } \Omega \\ p = y & \text{on } \Gamma_b \\ \frac{\partial p}{\partial n} = 0 & \text{on } \Gamma_d \cup \Gamma_a \\ \frac{\partial p}{\partial n} = \frac{q}{K} n_x & \text{on } \Gamma_f \quad (Neumann) \\ p = y & \text{on } \Gamma_f \quad (Dirichlet) \end{array} \right.$$

where the input water flux is $q = 0.02$, and $K = 0.5$. The velocity u of the water is given by $u = -\nabla p$.

We use the following fix point method : (with bad main B.C. *Execute freeboundaryPB.edp*) let be, $k = 0$, $\Omega^k = \Omega$. First step, we forgot the Neumann BC and we solve the problem : Find p in $V = H^1(\Omega^k)$, such $p = y$ on Γ_b^k et on Γ_f^k

$$\int_{\Omega^k} \nabla p \nabla p' = 0, \quad \forall p' \in V \text{ with } p' = 0 \text{ on } \Gamma_b^k \cup \Gamma_f^k$$

With the **residual of the Neumann boundary condition** we build a domain transformation $\mathcal{F}(x, y) = [x, y - v(x)]$ where v is solution of : $v \in V$, such than $v = 0$ on Γ_a^k (bottom)

$$\int_{\Omega^k} \nabla v \nabla v' = \int_{\Gamma_f^k} \left(\frac{\partial p}{\partial n} - \frac{q}{K} n_x \right) v', \quad \forall v' \in V \text{ with } v' = 0 \text{ sur } \Gamma_a^k$$

remark : we can use the previous equation to evaluate

$$\int_{\Gamma_f^k} \frac{\partial p}{\partial n} v' = - \int_{\Omega^k} \nabla p \nabla v'$$

Implementation

The new domain is : $\Omega^{k+1} = \mathcal{F}(\Omega^k)$ Warning if is the movement is too large we can have triangle overlapping.

```
problem Pp(p,pp,solver=CG) =
    int2d(Th) ( dx(p)*dx(pp)+dy(p)*dy(pp) )
+ on(b,f,p=y) ;
problem Pv(v,vv,solver=CG) =
    int2d(Th) ( dx(v)*dx(vv)+dy(v)*dy(vv) )
+ on(a, v=0)
+ int1d(Th,f) (vv*
    ((Q/K)*N.y-(dx(p)*N.x+dy(p)*N.y)) ) ;
while(errv>1e-6)
{
    j++; Pp; Pv;    errv=int1d(Th,f) (v*v) ;
    coef = 1;
    //      Here french cooking if overlapping see the example
    Th=movemesh(Th,[x,y-coef*v]) ;           //      deformation
}
Execute freeboundary.edp
```

Bose Einstein Condensate

Just a direct use of Ipopt interface (2day of works)

The problem is find a complex field u on domain \mathcal{D} such that :

$$u = \underset{\|u\|=1}{\operatorname{argmin}} \int_{\mathcal{D}} \frac{1}{2} |\nabla u|^2 + V_{trap} |u|^2 + \frac{g}{2} |u|^4 - \Omega i \bar{u} \left(\left(\frac{-y}{x} \right) \cdot \nabla \right) u$$

to code that in FreeFem++

use

- Ipopt interface (<https://projects.coin-or.org/Ipopt>)
- Adaptation de maillage

Run: BEC.edp

Hyper elasticity equation

The Hyper elasticity problem is the minimization of the energy $W(I_1, I_2, I_3)$ where I_1, I_2, I_3 are the 3 invariants. For example The Ciarlet Geymonat energy model is

$$W = \kappa_1(J_1 - 3) + \kappa_2(J_2 - 3) + \kappa(J - 1) - \kappa \ln(J)$$

where $J_1 = I_1 I_3^{-\frac{1}{3}}$, $J_2 = I_2 I_3^{-\frac{2}{3}}$, $J = I_3^{\frac{1}{2}}$,

let u the displacement, when

- $F = I_d + \nabla u$
- $C = {}^t F F$
- $I_1 = \text{tr}(C)$
- $I_2 = \frac{1}{2}(\text{tr}(C)^2 - \text{tr}(C^2))$
- $I_3 = \det(C)$

The problem is find

$$u = \underset{u}{\operatorname{argmin}} W(I_1, I_2, I_3)$$

Hyper elasticity equation

```
fespace Wh(Th, [P2,P2]);  
  
//      methode de Newton ..  
  
Wh [d1,d2]=[0,0];  
Wh [w1,w2],[v1,v2];  
for(int i=0;i<Nnewton;++i)  
{  
  solve dWW([w1,w2],[v1,v2]) =  
    int2d(Th) ( ddW2d([d1,d2],[w1,w2],[v1,v2]) )  
    - int2d(Th) ( dW2d([d1,d2],[v1,v2]) - [v1,v2]' * [f1,f2] )  
    + on(1,w1=0,w2=0);  
  
  d1[] -= w1[];  
  real err = w1[].linf;ty;  
  if(err< epsNewton) break;  
}
```

Run:Hyper-Elasticity-2d.edp

Run:ElasticLaw2d.edp

Run:CiarletGemoni.edp

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools
- 5 Schwarz method with overlap
- 6 Advance exercice
- 7 No Linear Problem
- 8 Technical Remark on freefem++**

compilation process on Windows

- ❶ **Download and install MINGW32 see**
<http://sourceforge.net/projects/mingw/files/Installer/mingw-get-inst/mingw-get-inst-20120426/>
- ❷ **Under mingw32 install wget and unzip**
 - `mingw-get install msys-wget`
 - `mingw-get.exe install msys-unzip`
- ❸ **To install freeglut of win32 for the graphics part**

```
wget http://files.transmissionzero.co.uk/software/development/GLUT/freeglut-MinGW.zip
unzip freeglut-MinGW-2.8.0-1.mp.zip
cp freeglut/include/* /c/MinGW/include/GL/.
cp freeglut/lib*.a /c/MinGW/lib/.
cp freeglut/freeglut.dll /bin
```
- ❹ **install a good blas (OpenBlas)** <http://xianyi.github.com/OpenBLAS/>
- ❺ **install MPI for // version HPC Pack 2008 SDK and HPC Pack 2008 R2 Service Pack 2**
- ❻ **install inno setup to build installer :** <http://www.xs4all.nl/~mלאan2/ispack/isetup-5.4.0.exe>
- ❼ **GSL for gsl interface** <http://sourceforge.net/projects/mingw-cross/files/%5BLIB%5D%20GSL/mingw32-gsl-1.14-1/mingw32-gsl-1.14-1.zip/download>

Finally, the configure argument are :

```
./configure '--enable-download' 'FC=mingw32-gfortran' 'F77=mingw32-gfortran' 'CC=mingw32-gcc'
'CXX=mingw32-g++' '-with-blas=/home/hecht/blas-x86/libgoto2.dll' 'CXXFLAGS=-I/home/hecht/blas-x86'
'--enable-generic' '--with-wget=wget' 'MPIRUN=/c/Program Files/Microsoft HPC Pack 2008 R2/Bin/mpiexec.exe'
```

Dynamics Load facility

Or How to add your C++ function in FreeFem++.

First, like in cooking, the first true difficulty is how to use the kitchen.

I suppose you can compile the first example for the `examples++-load`

```
numermac11:FH-Seville hecht# ff-c++ myppm2rnm.cpp
```

```
...
```

```
add tools to read pgm image
```

The interesting code

```
#include "ff++.hpp"
typedef KNM<double> * pRnm;
typedef KN<double> * pRn;
typedef string ** string;

pRnm read_image( pRnm const & a,const pstring & b);

pRn seta( pRn const & a,const pRnm & b)
{ *a=*b;
  KN<double> aa=*a;
  return a;}

void Init(){
// add ff++ operator "<-" constructor of real[int,int] form a string
TheOperators->Add("<-",
  new OneOperator2_<KNM<double> *,KNM<double> *,string*>(&read_image) );
// add ff++ an affection "=" of real[int] form a real[int,int]
TheOperators->Add("=",
  new OneOperator2_<KN<double> *,KN<double> *,KNM<double>* >(seta));
}
LOADFUNC(Init); // to call Init Function at load time
```

Remark, **TheOperators** is the ff++ variable to store all world operator, **Global** is to store function.

How to extend

A true simple example How to make dynamic gnuplot

Idea : use a pipe to speak with gnuplot the C code :

```
FILE * gp = popen("gnuplot");  
for( double f=0; f < 3.14; f += 0.01)  
    fprintf(gp, "plot sin(x+%f)\n", f);
```

To do this add a new constructor of ofstream in freefem++

A way to pass info between to code

Make a pipe, under unix (with a use of pstream tools)

```
#include "ff++.hpp"
#include "pstream.h"
typedef redi::pstream pstream;
typedef std::string string;
static pstream ** pstream_init(pstream **const & p, string * const & a)
{ *p = new pstream(a->c_str());
  return p;};

void initttt()
{
    // add new pointer type * pstream
    Dcl_TypeandPtr<pstream*>(0,0,::InitializePtr<pstream*>,::DeletePtr<pstream*>);
    // add cast operation to make std iostream read and write
    atype<istream* >()->AddCast( new E_Fl_funcT<istream*,pstream*>(UnRef<istream* >));
    atype<ostream* >()->AddCast( new E_Fl_funcT<ostream*,pstream*>(UnRef<ostream* >));
    // the constructor from a string .
    TheOperators->Add("<-" ,new OneOperator2_<pstream**,pstream**,string*>(pstream_init) );
    // add new keyword type pstream
    zzzfff->Add("pstream",atype< pstream ** >());
}
LOADFUNC(initttt);
t

MBP-FH:plugin hecht$ ff-c++ pipe.cpp
/usr/local/bin/g++ -c -g -m64 -fPIC -DNDEBUG -O3 -DBAMG_LONG_LONG -DNCHECKPTR -fPIC -I/usr/local/lib/ff++/3.20/include
/usr/local/bin/g++ -bundle -undefined dynamic_lookup -g -m64 -fPIC -DNDEBUG -O3 -DBAMG_LONG_LONG -DNCHECKPTR -fPIC 'pi
```

a small test : [Run:gnuplot.edp](#)

FreeFem++ et C++ type

The tools to add a operator with 2 arguments :

```
OneOperator2_<returntype ,typearg1 ,typearg2>(& thefunction );  
returntype thefunction(typearg1 const &, typearg2 const &)
```

To get the C++ type of all `freefem++` type, method, operator, just do in `examples++-tutorialdirectory`

```
c++filt -t < lestable  
Cmatrix 293 Matrice_Creuse<std::complex<double> >  
R3 293 Fem2D::R3  
bool 293 bool*  
complex 293 std::complex<double>*  
element 293 (anonymous namespace)::lgElement  
func 294 C_F0  
    ifstream 293 std::basic_istream<char, std::char_traits<char> >*<br>  
int 293 long*  
matrix 293 Matrice_Creuse<double>  
mesh 293 Fem2D::Mesh**  
mesh3 293 Fem2D::Mesh3**  
ofstream 293 std::basic_ostream<char, std::char_traits<char> >*<br>  
problem 294 Problem  
real 293 double*  
solve 294 Solve  
string 293 std::basic_string<char, std::char_traits<char>, std::allocator<char> >*<br>  
varf 294 C_args  
vertex 293 (anonymous namespace)::lgVertex
```

FreeFem++ Triangle/Tet capability

```
Element::nv;
const Element::Vertex & V = T[i];
double a = T.measure();
Rd AB = T.Edge(2);
Rd hC = T.H(2);
R l = T.lenEdge(i);
(Label) T ;
R2 G(T(R2(1./3,1./3)));
```

// soit T un Element de sommets $A, B, C \in \mathbb{R}^2$
// -----
// number of vertices of triangle (here 3)
// the vertex i of T ($i \in 0, 1, 2$)
// measure of T
// edge vector
// gradient of 2 base fonction
// length of i edge oppose of i
// label of T (region number)
// The barycentre of T in 3d

FreeFem++ Mesh/Mesh3 capability

```
Mesh Th("filename");           // read the mesh in "filename"
Th.nt;                          // number of element (triangle or tet)
Th.nv;                          // number of vertices
Th.neb or Th.nbe;              // number of border element (2d) or (3d)
Th.area;                        // area of the domain (2d)
Th.peri;                        // length of the border
typedef Mesh::Rd Rd;           // R2 or R3
Mesh2::Element & K = Th[i];    // triangle i, int i ∈ [0, nt[
Rd A=K[0];                     // coord of vertex 0 of triangle K
Rd G=K(R2(1./3,1./3));         // the barycentre de K.
Rd DLambda[3];
K.Gradlambda(DLambda);        // compute the 3  $\nabla \lambda_i^K$  for i=0,1,2
Mesh::Vertex & V = Th(j);      // vertex j, int j ∈ [0, nv[
Mesh::BorderElement & BE=th.be(l); // border element l ∈ [0, nbe[
Rd B=BE[1];                   // coord of vertex 1 on Seg BE
Rd M=BE(0.5);                 // middle of BE.
int j = Th(i,k);              // global number of vertex k ∈ [0,3[ of tria. i ∈ [0, nt[
Mesh::Vertex & W=Th[i][k];     // vertex k ∈ [0,3[ of triangle i ∈ [0, nt[

int ii = Th(K);               // number of triangle K
int jj = Th(V);               // number of triangle V
int ll = Th(BE);              // number of Seg de bord BE
assert( i == ii && j == jj ); // check.
```

- 1 Introduction
- 2 Tools
- 3 Academic Examples
- 4 Numerics Tools
- 5 Schwarz method with overlap
- 6 Advance exercice
- 7 No Linear Problem
- 8 Technical Remark on freefem++

Phase change with Natural Convection

The starting point of the problem is Brainstorming session (part I) of the third FreeFem++ days in december 2011, this is almost the Orange Problem is describe in web page <http://www.ljll.math.upmc.fr/~hecht/ftp/ff++days/2011/Orange-problem.pdf>. The coupling of natural convection modeled by the Boussinesq approximation and liquid to solid phase change in $\Omega =]0, 1[^2$, No slip condition for the fluid are applied at the boundary and adiabatic condition on upper and lower boundary and given temperature θ_r (resp θ_l) at the right and left boundaries.

The model is : find the field : the velocity $\mathbf{u} = (u_1, u_2)$, the pressure p and temperature θ :

$$\left\{ \begin{array}{lll} \mathbf{u} & \text{given} & \text{in } \Omega_s \\ \partial_t \mathbf{u} + (\mathbf{u} \nabla) \mathbf{u} + \nabla \cdot \mu \nabla \mathbf{u} + \nabla p & = -c_T \mathbf{e}_2 & \text{in } \Omega_f \\ \nabla \cdot \mathbf{u} & = 0 & \text{in } \Omega_f \\ \partial_t \theta + (\mathbf{u} \nabla) \theta + \nabla \cdot k_T \nabla \theta & = \partial_t S(T) & \text{in } \Omega \end{array} \right. \quad (12)$$

Where Ω_f is the fluid domain and the solid domain is $\Omega_s = \Omega \setminus \Omega_f$.

Phase change with Natural Convection

The enthalpy of the change of phase is given by the function S ; μ is the relative viscosity, k_T the thermal diffusivity.

In $\Omega_f = \{x \in \Omega; \theta > \theta_f\}$, with θ_m the melting temperature the solid has melt.











We modeled, the solid phase as a fluid with huge viscosity, so :

$$\mu = \begin{cases} \theta < \theta_f & \sim 10^6 \\ \theta \geq \theta_m & \sim \frac{1}{\text{Re}} \end{cases},$$

The Stefan enthalpy S_c with defined by $S_c(\theta) = H(\theta)/S_{th}$ where S_{the} is the stefan number, and H is the Heaviside function with use the following smooth the enthalpy :

$$S(\theta) = \frac{\tanh(50(\theta - \theta_m))}{2S_{te}}.$$

The true device



We apply a fixed point algorithm for the phase change part (the domain Ω_f is fixed at each iteration) and a full no-linear Euler implicit scheme with a fixed domain for the rest. We use a Newton method to solve the non-linearity.

- if we don't make mesh adaptation, the Newton method do not converge
- if we use explicit method diverge too,
- if we implicit the dependance in Ω_s the method also diverge.

This is a really difficult problem.

The finite element space to approximate u_1, u_2, p, θ is defined by

```
fespace Wh (Th, [P2,P2,P1,P1]) ;
```

We do mesh adaptation a each time step, with the following code :

```
Ph ph = S(T), pph=S(Tp);  
Th= adaptmesh (Th, T, Tp, ph, pph, [u1,u2], err=errh,  
                hmax=hmax, hmin=hmax/100, ratio = 1.2);
```

This mean, we adapt with all variable plus the 2 melting phase a time $n + 1$ and n and we smooth the metric with a ratio of 1.2 to account for the movement of the melting front.

The Newton loop

the fixed point are implemented as follows

```
real err=1e100,errp ;
for(int kk=0;kk<2;++kk) // 2 step of fixe point on  $\Omega_s$ 
{ nu = nuT; // recompute the viscosity in  $\Omega_s, \Omega_f$ 
  for(int niter=0;niter<20; ++ niter) // newton loop
  { BoussinesqNL;
    err = ulw[].linfo;
    cout << niter << "_err_NL_" << err << endl;
    ul[] -= ulw[];
    if(err < tolNewton) break; } // convergence ..
}
```

The linearized problem

```
problem BoussinesqNL([u1w,u2w,pw,Tw],[v1,v2,q,TT])
= int2d(Th) (
    [u1w,u2w,Tw]'*[v1,v2,TT]*cdt
  + UgradV(u1,u2,u1w,u2w,Tw)' * [v1,v2,TT]
  + UgradV(u1w,u2w,u1,u2,T)' * [v1,v2,TT]
  + ( Grad(u1w,u2w)'*Grad(v1,v2)) * nu
  + ( Grad(u1,u2)'*Grad(v1,v2)) * dnu* Tw
  + cmT*Tw*v2 + grad(Tw)'*grad(TT)*kT
  - div(u1w,u2w)*q -div(v1,v2)*pw - eps*pw*q
  + dS(T)*Tw*TT*cdt )
- int2d(Th) (
    [u1,u2,T]'*[v1,v2,TT]*cdt
  + UgradV(u1,u2,u1,u2,T)' * [v1,v2,TT]
  + ( Grad(u1,u2)'*Grad(v1,v2)) * nu
  + cmT*T*v2 - eps*p*q + grad(T)'*grad(TT)*kT
  - div(u1,u2)*q -div(v1,v2)*p
  + S(T)*TT*cdt - [u1p,u2p,Tp]'*[v1,v2,TT]*cdt
  - S(Tp)*cdt*TT)
+ on(1,2,3,4, u1w=0,u2w=0)+on(2,Tw=0)+on(4,Tw=0) ;
```



The parameters of the computation

take case 2 from

Shimin Wang, Amir Faghri, and Theodore L. Bergman. A comprehensive numerical model for melting with natural convection. *International Journal of Heat and Mass Transfer*, January 2010.

$\theta_m = 0$, $Re = 1$, $S_{te} = 0.045$, $P_r = 56.2$, $R_a = 3.27 \cdot 10^5$, $\theta_l = 1$, $\theta_r = -0.1$ so in this case $cmT = c_T = -R_a/P_r$, $kT = k_T = 1/P_r$, $eps = 10^{-6}$, time step $\delta t = 10^{-1}$, $cdt = 1/\delta t$, at time $t = 80$ and we get a good agreement with the article.

Phase change with Natural Convection



So now, a real problem, get the physical parameter of the real experiment.

Run:Orange-Newton.edp

Freefem++ v3.23 is

- very good tool to solve non standard PDE in 2D/3D
- to try new domain decomposition domain algorithm

The the future we try to do :

- Build more graphic with VTK, paraview , ... (in progress)
- Add Finite volume facility for hyperbolic PDE (just begin C.F. FreeVol Projet)
- 3d anisotrope mesh adaptation
- automate the parallel tool

Thank for you attention.